Solids of Revolution

```
> restart;
```

Generating a Solid by Rotating a Plane Curve about the x -axis

```
> with(plots):
with(plottools):
```

The following procedure will create an animation of a curve of the form $y = f(x)$, x in [a, b], being rotated about the x -axis.

The parameters are

f , the function defining the curve to be rotated,

a , b , the interval on the x -axis over which to create the animation,

n , the number of frames used in the animation.

```
> RotateX := proc(f, a, b, n)
local k, theta, g, R, c, L, A, B, arc, T;
for k from 1 to n do
theta := k*2*Pi/n:
g[k] := plot3d([x, f(x)*cos(t), f(x)*sin(t)], x=a..b, t=0..theta):
R[k] := spacecurve([b, f(b)*cos(theta)*t, f(b)*sin(theta)*t], t=0..1, color=red):
c[k] := spacecurve([b, 0.25*cos(t), 0.25*sin(t)], t=0..theta, color=red):
od:
L := line([b,0,0],[b,1,0],color=red):
g[0] := spacecurve([x, f(x), 0], x=a..b, thickness=2):
R[0] := line([b,f(b),0], [b,0,0], color=red):
c[0] := L:
A := display([seq(g[k], k=0..n)], insequence=true, axes=normal, orientation=[-58,29], style=patch):
B := display([seq(R[k], k=0..n)], insequence=true, axes=normal, orientation=[-58,29]):
arc := display([seq(c[k], k=1..n)], insequence=true, axes=normal, orientation=[-58,29]):
T := textplot3d([b, 0.4*cos(Pi/8), 0.4*sin(Pi/8), `q`], font=[SYMBOL,10], color=red):
display([A, B, arc, T, L]);
end:
```

Here is an example using the function $y = \frac{1}{x}$ on the interval [1, 5] with 15 frames.

Click on the graph, and run the animation.

```
> n := 15:
f := x -> 1/x:
a := 1: b:=5:
RotateX(f, a, b, n);
```



An Animation of the Disk Method

The following procedure creates an animation of n disks being stacked to illustrate the formula

$$V = \int_{c}^{d} \pi\, \mathrm{f}(y)^2\, dy \quad ,$$

which computes the volume of the solid obtained by rotating the curve $x = \mathrm{f}(y)$ about the y -axis.

The parameters are

f , the function,

c, d , the interval over which the disks are being stacked,

n , the number of disks displayed (keep this parameter modest).

```
> DiskY := proc(f, c, d, n)
local h, k;
h:=(d-c)/(n-1):
for k from 1 to n do
```

```
disk.k := cylinder([0,0,c+(k-1)*h], f(c+(k-1)*h), h):
p.k := plots[display]([seq(disk.j, j=1..k)], scaling=constrained, style=patchnogrid, axes=framed,
orientation=[45,65]):
od:
plots[display]([seq(p.k, k=1..n)], insequence=true);
end:
```

Here is an example using $x = \frac{2}{y}$ on [1, 4] using 6 disks.

Click on the graph and run the animation.

```
> n := 6:
c:=1: d:=4:
f := y->2/y;
DiskY(f, c, d, n);
```

$$f := y \to 2\,\frac{1}{y}$$



Assignment 13

Ex.1:

Consider the function $f(x) = \arctan\left(\frac{x-1}{x+1}\right)$ for x in $[-\frac{1}{2}, 4]$.

a) Use **RotateX** to create an animation of the solid of revolution obtained by rotating the graph of f

around the x -axis.

b) Use the procedure **DiskY** to create an illustration of the disk method for the function $f = f(y)$, with y in $[-\frac{1}{2}, 4]$.

Ex.2:

An embankment is to be built around a circular wading pool as shown in the figure below.

The dimensions are as follows:

the inner radius (of the blue area) is $5\,m$,

the width of the embankment (the brown area) is $6\,m$.

```
> restart;
with(plots):
p1:=plot3d([x*cos(t), x*sin(t), (11-x)*(x-5)^2/16], t=0..2*Pi, x=5..11, orientation=[45,65],
color=brown, style=wireframe, scaling=constrained):
p2:=plot3d([x*cos(t), x*sin(t), 0], t=0..2*Pi, x=0..5, color=blue, style=patchnogrid):
display([p1,p2]);
```



The next figure shows the profile of the embankment.

It is $6\,m$ wide, and $2\,m$ high at a distance $2\,m$ from the outer edge.

```
> plot((11-x)*(x-5)^2/16, x=5..11, color=brown, scaling=constrained);
```



a) Find a cubic polynomial p which fits the three points ($5, 0$), ($9, 2$), and ($11, 0$), and has a gentle slope at the inside, i.e., satisfies $\mathrm{D}(p)(5) = 0$.

Hint: Set up p as a general cubic polynomial, and determine its coefficients by solving the 4 equations $\mathrm{p}(x_i) = y_i$, $i = 1, 2$, 3, and $\mathrm{D}(p)(5) = 0$.

b) Determine the amount of fill required to build the embankment.

Ex.3:

Consider the curves $y_1 = \frac{x}{3x^2 + 1}$ and $y_2 = m\,x$.

a) For what values of m do y_1 and y_2 bound an area?

b) Create a plot of the curves for some admissible value of m .

c) Find the enclosed area (as a function of m).

d) Create a plot that shows the dependence of the area of the enclosed region on m .

Is there a value of m for which this area is maximal? Comment.

e) How should m be chosen such that the enclosed area equals 1000?